SEM Power Analyses

Aki hamagami

Feb. 16, 2006

Definition of Power Analyses

- Power: probability of being able to reject H_0 when H_0 is false.

General steps for Power Analysis

- Establish H_0
- Specify test statistics
- Establish a critical value for cumulative distribution function of test statistics, or rejection region
- Specify H_1 or alternative model (hypothesis)
- Obtain test statistics based on H_1
- Calculate Power by comparing central and non-central distributions of test statistics

Under SEM ML estimation

$$F_{ML} = \ln\left|\hat{\Sigma}\right| - \ln\left|S\right| + Tr\left(S\hat{\Sigma}^{-1}\right) - p$$

- Given the model, df, N=n+1

  H_0: $F_0 = 0$ (the model is perfect fit)
- Sampling test statistic nF_{ML}
- Distribution under H_0: $nF_{ML} \sim \chi^2(df)$

  where df=p(p+1)/2 - # of parameters
- Under H_0, obtain critical value of χ^2_c
- Under the alternative model, specify H_1:

  H_1: $F_1 > 0$ (the model is not a perfect)

Continued

- Distribution under H_1: $nF_1 \sim nc\chi^2(df, \lambda)$

Noncentrality parameter $\lambda = nF_1$

- Power is AREA under the noncentral distribution beyond critical value of χ^2_c

Relationship between the true distribution and false distribution



$$P = \Pr\left(\chi^2(df, \lambda) > \chi_c^2\right)$$

Mathematical formula for central χ^2 density function

$$P = \int_{\chi_c^2}^{\infty} \frac{1}{2\Gamma(d/2)} \left(\frac{z}{2}\right)^{\frac{d}{2}-1} e^{-z/2} dz$$

Mathematical formula for noncentral χ^2 density function

$$P_r(x) = \frac{e^{-(x+\lambda)/2}\, x^{r/2-1}}{2^{r/2}} \sum_{k=0}^{\infty} \frac{(\lambda x)^k}{2^{2k}\, k!\, \Gamma\left(k + \frac{1}{2} r\right)}$$

$$\frac{e^{-(x+\lambda)/2}\, x^{(r-1)/2} \sqrt{\lambda}}{2\, (\lambda x)^{r/4}}\, I_{r/2-1}\left(\sqrt{\lambda x}\right)$$

$$2^{-r/2}\, e^{-(\lambda+x)/2}\, x^{r/2-1}\, {}_0\tilde{F}_1\left(; \frac{1}{2} r; \frac{1}{4} \lambda x\right),$$

Where I(x) is a modified bessel function, and F(*) is a confluent hypergeomtric Function.

Satorra & Saris Approach

- H0: the model fits perfectly while H1: the model does not fit perfectly.
- Specify the reference model and determine the parameters of the model
- Obtain the expected covariance matrix (Σ_0)based on the reference model
- Fit the alternative model to the expected covariance matrix
- Obtain the fit function index and use it as non-centrality parameter
- Compute the probability of rejecting the null hypothesis: F0=0 (the alternative model fits perfectly)

Satorra & Saris (1985) Example: Reference Model



Satorra & Saris (1985) Example: Alternative Model

Question: Does this model fit perfectly to the data?



$$Power = \Pr\left(\chi^2(df = 4, \lambda = 5.14) > \chi_c^2\right) = .41$$

Satorra & Saris (1985) Example: SAS Script

```
title 'Satorra & Saris Example';
data temp;
    d=4;
    * degree of freedom;
    alpha=.05;
    * alpha level;
    ncpa=5.14;
    * Noncentrality parameter from the alternative model;
    cval=cinv(1-alpha,d) ;
    * critical chi-square value at alpha level;
    power=1-probchi(cval,d,ncpa) ;
    * compute power;
run;
proc print data=temp;
run;
```

Satorra, A., & Saris, W. (1985). Power of the likelihood ratio test in covariance structure analysis. *Psychometrika, 5l,* 83-90.

Saris, W. E., & Satorra, A. (1993). Power evaluations in structural equation models. In K. A. Bollen & J. S. Long (Eds.)(pp.19-23), *Testing structural equation models.* Newbury Park, CA: Sage.

MacCallum, Browne, & Sugawara (1996) based on RMSEA

- **Test of Close Fit: H0: $\varepsilon \leq .05$**
- If the true value of ε is actually .08 and we test hypothesis that model fit is close, we ask the question: what is the likelihood of rejecting the null hypothesis?
- **Test of NOT-Close Fit: H0: $\varepsilon \geq .05$**
- If the model fit is actually good and we test the hypothesis that the fit is not even close, what is the likelihood of rejecting this hypothesis?

MacCallum, Browne, & Sugawara

- Use overall misfit index given ML, RMSEA

$$\varepsilon = \sqrt{\frac{F_0}{df}}$$

- A test of close fit

$$H0 : \varepsilon \le .05$$

- Under this H0, the test statistic nF0 follows a non-central χ^2 distribution with d degree of freedom and noncentrality parameter

$$\lambda 0 = nd\varepsilon^2 = nd(.05)^2$$

- A critical value of χ^2 is determined from that H0 Noncentral χ^2 distribution and the observed value of the test statistic is compared to this critical value

- Under the alternative hypothesis, the test statistic nFa is distributed as noncentral χ^2 with d degree of freedom and noncentral parameter:

$$\lambda 1 = nd(.10)^2$$

- Power then is a probability of rejecting H0: $\varepsilon \leq .05$ if in reality $\varepsilon = .10$

$$Power = P\left(\chi^2(d, \lambda 1) \geq \chi_c^2\right)$$

Power for the null hypothesis of CLOSE fit



Power as a function of RMSEA for alternative hypothesis







Power of rejecting the null hypothesis of NOT-CLOSE fit



Important equations under MacCallum et al approach

$$\varepsilon = \sqrt{\frac{F_0}{df}}$$

RMSEA

$$F_0 = df \times \varepsilon^2$$

ML fit function

$$\lambda = (N-1) \times F_0$$

$$= (N-1) \times df \times \varepsilon^2$$

Noncentrality Parameter

```
title "sas program for power estimation for SEM" ;
data one ;
alpha=.05 ;  *significance level ;
rmsea0=.05 ; *null hypothesis value ;
rmseaa=.10 ; *alt hypothesis value ;
d=50 ; *degrees of freedom ;
n=200 ; *sample size ;
   ncp0=(n-1)*d*rmsea0**2 ;
   ncpa=(n-1)*d*rmseaa**2 ;
   * Test for Close Fit;
   if rmsea0 lt rmseaa then do ;
              cval=cinv(1-alpha,d,ncp0) ;
              power=1-probchi(cval,d,ncpa) ;
   end ;
  * Test for Not-Close Fit;
   if rmsea0 gt rmseaa then do ;
              cval=cinv(alpha,d,ncp0) ;
              power=probchi(cval,d,ncpa) ;
   end ;
   output ;
proc print data=one ; var rmsea0 rmseaa alpha d n power ; run
```

Power Analyses of Test of Exact Fit between two competing alternative models

- A test of Exact fit (No difference)

$$H0: \Delta\varepsilon = .00$$

$$H0: \Delta\chi^2 = 0$$

- Power: What is probability of rejecting H0 of zero $\Delta\chi 2$ when it is not zero at all?
- Under this H0, the test statistic $\Delta\chi 2$ follows a non-central $\chi 2$ distribution with Δd degree of freedom and noncentral parameter

$$\lambda 0 = nd(\Delta\varepsilon)^2 = \Delta\chi^2$$

$$Power = P\left(\chi^2(\Delta d, \Delta\lambda) \geq \chi_c^2\right)$$



Examples of Power Analyses of Test of Exact Fit between two competing alternative models

- A model of interest: Linear growth model: M0
- An alternative model: No growth model: M1
- H0: $\Delta\chi 2=0$ meaning that M0 and M1 fit the data exactly the same
- A sample fit statistic: $\Delta\chi 2=(N-1)*(F_{m1}-F_{m0})=20$ with $\Delta df=5$
- What is a power of rejecting the H0 when it is not true?

SAS script to compute power

```
data temp;
   d=5; * degree of freedom;
   alpha=.05;  * alpha level;
   ncpa=20;  * Noncentrality parameter from analyses;
   cval=cinv(1-alpha,d) ;  * critical chi-square value at alpha level;
   power=1-probchi(cval,d,ncpa) ; * compute power;
run;
proc print data=temp;
run;
```

$$Power = P\left(\chi^2(\Delta d = 5, \Delta\lambda = 20) \geq \chi_c^2\right) = .95$$

Matlab script to compute power

```
% Set up parameters
alpha=.05
df=5
noncentrality=20

% Compute critical chi-square
CriticalChi2=chi2inv(1-alpha, df)

% Compute power
beta=ncx2cdf(CriticalChi2, df, noncentrality)
power=1-beta
```

CriticalChi2 =   11.0705
beta = 0.0477
power = 0.9523

Statistical Power as a function of Noncentrality parameter



The larger the df, the larger Noncentrality parameter is needed to achieve The same power. The parsimonious model contains less number of parameter and more df than complex models.